PSPICE Command Summary

Below are found brief descriptions regarding PSPICE commands and syntax. The descriptions are in no way complete and for more information one must refer to PSPICE documentation and books. The descriptions below use the following conventions:

`|name|`	means required in definition
`{name}`	means optional in definition
`[a,b,c]`	means choice of a, b, or c in definition
`(name)`	means parentheses required in definition
`...`	means multiple declarations of the same format

TABLE A.1. PSPICE commands

Command	Description	Page
`*`	Line Comment	A1
`;`	In-line Comment	A2
`+`	Line Continuation	A2
`.AC`	AC Analysis	A2
`.DC`	DC Analysis	A2
`.END`	End of Circuit	A3
`.ENDS`	End of Sub-circuit	A3
`.FOUR`	Fourier Analysis	A4
`.IC`	Initial Transient Condition	A4
`.INC`	Include File	A4
`.LIB`	Library File	A4
`.MC`	Monte Carlo Analysis	A5
`.MODEL`	Model	A5
`.NODESET`	Node set	A6
`.NOISE`	Noise Analysis	A6
`.OP`	Bias Point Analysis	A7
`.OPTIONS`	Options	A7
`.PLOT`	Plot	A8
`.PRINT`	Print	A8
`.PROBE`	Probe	A9
`.SENS`	Sensitivity Analysis	A9
`.SUBCKT`	Sub-circuit Definition	A9
`.TEMP`	Temperature	A10

Command	Description	Page
`.TF`	Transfer Function	A10
`.TRAN`	Transient Analysis	A10
`.WIDTH`	Width	A11
`B`	GaAsFET	A11
`C`	Capacitor	A12
`D`	Diode	A12
`E`	Voltage Controlled Voltage Source	A13
`F`	Current Controlled Current Source	A14
`G`	Voltage Controlled Current Source	A14
`H`	Current Controlled Current Source	A15
`I`	Independent Current Source	A15
`J`	Junction FET	A16
`K`	Inductor Coupling (Transformer Core)	A17
`L`	Inductor	A18
`M`	MOSFET	A18
`Q`	Bipolar Transistor	A20
`R`	Resistor	A21
`S`	Voltage Controlled Switch	A21
`T`	Transmission Line	A22
`V`	Independent Voltage Source	A22
`W`	Current Controlled Switch	A23
`X`	Sub-circuit Call	A24
`SOURCES`	Transient Source Descriptions	A24

* - Line Comment

Lines that start with a * are comment lines. PSPICE reads them in but does not execute the line.

Example:

```
*     This line is a comment
```

; - In-line Comment

When a line has a **;** in it, PSPICE treats everything to the right of the **;** as a comment.

Example:

```
C1     1     0     1P     ;Comment now
```

+ - Continuation of Previous Line

When a line begins with a **+** PSPICE regards the line as a continuation of the line above it.

Example:

```
.MODEL MNPN NPN IS=1e-15 BF=100 RE=5
+ RB=50 CJE=10f
```

Both lines are considered to describe the model **MNPN**.

.AC - AC (Frequency) Analysis

General Format:

```
.AC  [LIN,DEC,or OCT] |# points| |start frequency| |stop frequency|
```

.AC tells PSPICE to run the circuit of the frequency range given. The frequency range is from the **|start frequency|** to the **|stop frequency|** inclusive, with the circuit simulated **|# points|** times per division. The division is specified by **LIN**, **DEC**, or **OCT**.

LIN tells it to run **|# points|** simulations over the entire range of frequencies.
DEC tells it to run **|# points|** simulations per decade (power of 10) over the entire range of frequencies.
OCT tells it to run **|# points|** simulations per octave (power of 2) over the entire range of frequencies.

Example:

```
.AC  DEC  20  1  1MEG
```

PSPICE runs an AC analysis over the range of 1 Hz to 1 MHz with 20 simulation points per decade.

.DC - DC Analysis

General Formats (multiple ways of declaration):

```
.DC [LIN] |sweep variable name| |start value| |end value|
+         |increment value| {nested sweep}
```

or

```
.DC [OCT,DEC] |sweep variable name| |start value| |end value|
+             |# points| {nested sweep}
```

or

```
.DC |sweep variable name| LIST |values ...| {nested sweep}
```

.DC tells PSPICE to find the DC bias point of the circuit while **|sweep variable name|** is changing values. The sweep can be nested (over multiple variables). The value is swept over the specified range and the number of simulations is determined by the **LIN**, **DEC**, **OCT**, or **LIST** declaration.

LIN tells it to run simulation over the specified range with **|increment value|** increments.
DEC tells it to run **|# points|** simulations per decade (power of 10) over the specified range.
OCT tells it to run **|# points|** simulations per octave (power of 2) over the specified range.
LIST runs simulations only over the values given in **|values ...|**

The **|sweep variable name|** can either be an independent source or a model parameter.

Examples:

.DC LIN VIN 0 5 1
sweeps source **VIN** from 0V to 5V in 1V increments.

.DC RES RMOD(R) 0.9 1.0 0.001
sweeps the resistance multiplier of the **RMOD** model from 0.9 to 1.0 in 0.01 increments.

.DC TEMP LIST -55 27 125
runs simulations at temperatures -55C, 27°C, and 125C.

.DC LIN VIN 3 5 1 VCC 4.9 5.1 0.01
runs a nested sweep with sources **VIN** and **VCC**. **VCC** is considered the "inner loop" while **VIN** is considered the "outer loop". Thus **VIN** is fixed, and **VCC** is swept linearly over the range of 4.9V to 5.1V in 0.01V increments. Then **VIN** is incremented and **VCC** is again swept over its range. This is continued over the entire **VIN** sweep range from 3V to 5V in 1V increments.

.END - End of Circuit

General Format:

.END

The **.END** declaration tells PSPICE its the end of the circuit description. All data and commands must be placed before the **.END** line. Multiple circuits can be run in the same net list, however, the circuits must be separated by the **.END** command.

Example:

.END

.ENDS - End of Sub-circuit

General Format:

.ENDS {Sub-circuit name}

The **.ENDS** command declares the end of the **.SUBCKT** definition. It is good practice to put the name of the Sub-circuit after the **.ENDS**, though it is not required.

Examples:

.ENDS
ends the **.SUBCKT** description

.ENDS OPAMP
ends the **.SUBCKT OPAMP** description.

.FOUR - Fourier Analysis

General Format:

.FOUR |frequency value| |output variable|

A Fourier analysis is performed and the **|output variable|** signal transient analysis results are decomposed into its Fourier components with the fundamental frequency given by **|frequency value|**.

Example:

```
.TRAN 1u 1m
.FOUR 10K V(5) V(7)
```

The results of the transient analysis run for the voltages on nodes 5 and 7 are decomposed into its Fourier components with 10kHz as the fundamental frequency.

.IC - Initial Transient Condition

General Format:

.IC |V(|node #|) = |value| ...|

.IC fixes the node voltage of node **|node #|** to **|value|** volts over the entire bias point analysis. Thus **.IC** tells PSPICE the node voltage at time **t=0** in the transient analysis. **.IC** is sometimes used to help the circuit easily converge (assuming you know the final value) or to set initial capacitor voltages. After the bias point analysis is completed the nodes are no longer held at the specified values and are able to change over time.

Example:

```
.IC V(5) = 0 V(7) = 5
```

The voltages at nodes 5 and 7 are set, respectively, to 0V and 5V for the duration of the bias point analysis.

.INC - Include File

General Format:

.INC |filename|

The **.INC** statement tells PSPICE to include the contents of **|filename|** in the analysis run. The included file cannot contain a **.END** statement.

Example:

```
.INC C:\LIB\DVCO.CIR
```

.LIB - Library File

General Format:

.LIB |filename|

The **.LIB** statement tells PSPICE to look for models in the file **|filename|**. Only the needed models are read in by PSPICE

Example:

```
.LIB MOT1.LIB
```

.MC - Monte Carlo (Statistical) Analysis

General Format:

```
.MC |# runs| [DC,AC, or TRAN] |output variable| YMAX
+   {LIST} {OUTPUT |output specification}
```

The **.MC** command tells PSPICE to run a Monte Carlo statistical analysis on the circuit looking at the **|output variable|** response based on variation of other parameters. The variable parameters are those parameters in the model which contain **DEV** and **LOT** tolerances (see **.MODEL** for more). The first run is done with nominal values of parameters. The **|# runs|** tells PSPICE how many runs it should do.

Only one of the **DC**, **AC**, or **TRAN** statements can be specified. **|output variable|** tells PSPICE to save the statistical information on the given variable. **YMAX** specifies the operation to be performed on the values of **|output variable|** to reduce these to a single value. This value is the basis for the comparisons between the nominal and subsequent runs.

LIST tells PSPICE to print out the model parameter values at the beginning of each run.

The output from the initial run is controlled by the **.PRINT**, **.PLOT**, or **.PROBE** commands in the net list. For the other runs the output is suppressed unless the **OUTPUT** keyword is present. Then the **|output specification|** tells what output is wanted. The **|output specification|** can be:

ALL	forces all output to be generated
FIRST |value|	gives output for the first **|value|** runs
EVERY |value|	gives output every **|value|** run
RUNS |value ...|	gives output for the runs listed in **|value ...|**

Examples:

.MC 10 TRAN V(5) YMAX
tells PSPICE to run 10 statistical transient runs saving V(5) values.

.MC 50 DC IC(Q7) YMAX LIST
tells PSPICE to run 50 statistical DC runs saving IC(Q7) values and listing the parameters for each run.

.MC 20 AC VP(13,5) YMAX LIST OUTPUT ALL
tells PSPICE to run 20 statistical AC runs saving VP(13,5) values, listing the parameters for each run, and outputting the results for each run.

.MODEL - Model Definition

General Format:

```
.MODEL |name| |type|
+     [|parameter name| = |value| {tolerance specfication} ...]
```

The **.MODEL** statement describes a set of device parameters which are used in the net list for certain components. **|name|** is the model name which the components used. **|type|** is the device type and must be one of the following:

CAP	capacitor	**PJF**	P-channel JFET
IND	inductor	**NMOS**	N-channel MOSFET
RES	resistor	**PMOS**	P-channel MOSFET
D	diode	**GASFET**	N-channel GaAs MESFET
NPN	NPN bipolar transistor	**CORE**	nonlinear magnetic core (transformers)
PNP	PNP bipolar transistor	**VSWITCH**	voltage controlled switch
NJF	N-channel JFET	**ISWITCH**	current controlled switch

Following **|type|** is the list of parameters which describe the model for the device. None, any, or all parameters may be assigned values, those that are not assigned take on default values. The lists of parameter names, meanings, and default values are located in the individual device descriptions.

|tolerance specification| is used by the .MC analysis and has the format:

```
[DEV |value|{%}][LOT |value|{%}]
```

LOT tolerances track so all devices that refer to the model have the same variation per run.
DEV tolerances are independent of each other, thus different devices will have different variations for each run. The % indicates a relative (percentage) tolerance. If omitted then **|value|** is in the same units as the parameter it describes.

Examples:

```
.MODEL MNPN NPN IS=1e-15 BF=100
```

describes a model **MNPN** which is a standard default **NPN** except with **IS=1e-15** and beta, **BF =100**.

```
.MODEL DLOAD D (IS=1e-9 DEV 0.5% LOT 10%)
```

describes a diode model **DLOAD** with given **IS** and variation over **LOT** and **DEV** given.

.NODESET - Nodeset

General Format:

```
.NODESET | V(|node|) = |value| ... |
```

The **.NODESET** gives PSPICE an initial guess for the DC bias point. PSPICE then iterates changing the node voltage specified until the circuit converges (DC bias point found). **.NODESET** is often used to speed up the time of convergence for the DC bias point solution.

Example:

```
.NODESET V(2)=1 V(5)=2
```

tells PSPICE that the initial guess for DC bias point calculations has V(2) = 1V and V(5) = 2V.

.NOISE - Noise Analysis

General Format:

```
.NOISE V(|node|{,|node|}) |name| {internal value}
```

The **.NOISE** specifies a noise analysis of the circuit. The noise analysis works with the **.AC** statement, thus the **.AC** statement must be included.

V(|node|{,|node|}) is an output voltage. It can either be the voltage at one node (e.g. V(5)) or the voltage difference between two nodes (e.g. V(4,5)).

|name| is not a noise generator, instead it is a place to calculate the equivalent input noise. **|name|** can be either a voltage source or a current source.

The noise-generating devices in the circuit are the resistors and semiconductor devices. For each AC frequency analysis, each noise generator's contribution is determined and propagated to the output nodes where they are all RMS summed. From this total value, and the gain of the circuit, the equivalent input noise is calculated.

{interval value} is optional and if included tells PSPICE the print interval. Every {interval value}th frequency, a table is printed detailing the noise contributions of each noise generator.
Example:

.NOISE V(5) VIN
tells PSPICE to run a **NOISE** analysis at output node 5 and refer it back to the input **VIN**.

.OP - Bias Point Analysis

General Format:

.OP

The **.OP** command tells PSPICE to print detailed information about the bias point in the **.OUT** file.

Example:

.OP

.OPTIONS - PSPICE Options

General Formats:

.OPTIONS |option name ...|

or

.OPTIONS (|option name|=|value| ...)

The **.OPTIONS** command is used to set all options, limits, and control parameters for the various analyses. There are two types of options: flags and value options. The flag options can just be listed, the value options need a specific value to be assigned to it.

TABLE A.2. Value Options:

Option	*Description*	*Assigned Value*
ABSTOL	best accuracy for currents	Units = amp ; default value = 1pA
CHGTOL	best accuracy for charges	Units = coulomb ; default value = 0.01pC
CPTIME	CPU time allowed for this run	Units = sec ; default value = 1E6
DEFAD	MOSFET default drain area (AD)	Units = $meter^2$; default value = 0
DEFAS	MOSFET default source area (AS)	Units = $meter^2$; default value = 0
DEFL	MOSFET default gate length (L)	Units = meter ; default value = 100u
DEFW	MOSFET default gate width (W)	Units = meter ; default value = 100u
GMIN	minimum conductance used for any branch	Units = ohm^{-1} ; default = 1E-12
ITL1	DC and bias point "blind" iteration limit	default value = 40
ITL2	DC and bias point "educated guess" iteration limit	default value = 20
ITL4	iteration limit at any point in transient analysis	default value = 10
ITL5	total iteration limit for all points in transient analysis (ITL5=0 means ITL5=infinity)	default value = 5000
LIMPTS	maximum # of points allowed for any print table or plot	default value = 201
NUMDGT	# of digits output in print tables (maximum 8 useful digits)	default value = 4
PIVREL	relative magnitude required for pivot in matrix solution	default value = 1E-3
PIVTOL	absolute magnitude required for pivot in matrix solution	default value = 1E-13
RELTOL	relative accuracy of V's and I's	default value = 0.001
TNOM	default temperature (also the temp at which model parameters are assumed to be measured)	Units = °C ; default value = 27
TRTOL	transient analysis accuracy adjustment	default value = 7.0
VNTOL	best accuracy of voltages	UNITS = Volts ; default value = 1uV

TABLE A.3. Flag Options:

Option	*Meaning*
ACCT	summary and accounting information is output at the end of all the analysis
LIST	summary of circuit devices is output
NODE	net list (node table) is output
NOECHO	suppresses listing of input file
NOMOD	suppresses listing of model parameters and temperature updated values
NOPAGE	suppresses paging and printing of a banner for each major section of output
OPTS	values for all options are output
WIDTH	same as **.WIDTH OUT =** statement

Examples:

.OPTIONS NOECHO
tells PSPICE not to repeat the input net list in the **.OUT** file

.OPTIONS LIMITS = 100
tells PSPICE to limits points in tables or plots to 100 points

.PLOT - Plot

General Formats:

```
.PLOT [DC,AC,NOISE,TRAN] |output variable ...|
+     {(|lower limit value|,|upper limit value|)}
```

The **.PLOT** command tells PSPICE to create a plot of the **|output variable|** signal in the **.OUT** file for either a **DC**, **AC**, **TRAN**, or **NOISE** analysis.

Examples:

.PLOT TRAN V(5) V(9)
tells PSPICE to make a plot of the results of a transient analysis run for the voltages at nodes 5 and 9.

.PLOT TRAN V(2,3) (0,5V)
tells PSPICE to create a plot of the results of a transient analysis run for the voltage difference between nodes 2 and 3. The plot is limited in range from 0V to 5V.

.PRINT - Print

General Formats:

```
.PRINT [DC,AC,NOISE,TRAN] |output variable ...|
```

The **.PRINT** statement tells PSPICE to create a table of the **|output variable|** signal in the **.OUT** file for either a **DC**, **AC**, **TRAN**, or **NOISE** analysis.

Examples:

.PRINT TRAN V(5) V(9)
tells PSPICE to make a table of the results of a transient analysis run for the voltages at nodes 5 and 9.

.PLOT TRAN V(2,3)
tells PSPICE to create a table of the results of a transient analysis run for the voltage difference between nodes 2 and 3.

.PROBE - Probe

General Formats:

```
.PROBE {|output variable ...|}
```

.PROBE tells PSPICE to write the results of the **DC**, **AC** or transient simulation in a format that the Probe graphics post-processor can read. If **.PROBE** is followed by the optional |output variable| then only those signals will be saved. **.PROBE** alone means save all signal data thus the file created (**PROBE.DAT**) may become huge.

Examples:

.PROBE
tells PSPICE to save all signal data into **PROBE.DAT**.

.PROBE V(2)
tells PSPICE to save only the signal voltage data for node 2 into **PROBE.DAT**.

.SENS - Sensitivity Analysis

General Formats:

```
.SENS |output variable ...|
```

.SENS tells PSPICE to run a **DC** sensitivity analysis. Once the circuit is linearized about the bias point, the sensitivities of each **|output variable|** to all device values and parameters are calculated and output. Be warned that output file can become huge!

Example:

.SENS V(1) I(VCC)
PSPICE will run a **DC** sensitivity analysis on node V(1) and I(VCC)

.SUBCKT - Sub-circuit description

General Formats:

```
.SUBCKT |name| {node ...}
```

.SUBCKT declares that a Sub-circuit of the net list will be described until the **.ENDS** command. Sub-circuits are called in the net list by the command, **X**. **|name|** is the Sub-circuits name. **{node ...}** is an optional list of nodes local only to the Sub-circuit and used for connection on the top level. Sub-circuit calls can be nested (can have **X** inside). However, Sub-circuits cannot be nested (no **.SUBCKT** inside).

Example:

```
.SUBCKT RES10 1 2 3
R 1 2 10
C1 1 3 1p
C2 2 3 1p
.ENDS
```

A Sub-circuit **RES10** is described as a resistor with parasitic capacitors whose body (or tub) can be hooked up to any node.

.TEMP - Temperature

General Formats:

```
.TEMP |value ...|
```

.TEMP tells PSPICE to run simulations at the listed temperatures. The listing is included in the **<value>** parameter.

Example:

```
.TEMP -55 27 125
```

PSPICE will simulate the circuit at T = -55°C, 27°C, and 125°C

.TF - Transfer Function

General Formats:

```
.TF |output variable| |input source|
```

.TF forces PSPICE to calculate the small signal transfer function of the circuit around the bias point. The gain from **|input source|** to **|output variable|**, input resistance, and output resistances are output.

Example:

```
.TF V(OUT) VIN
```

PSPICE will calculate the transfer function and input and output resistances from source **VIN** to node **V(OUT)**. All information is sent to the **.OUT** file.

.TRAN - Transient Analysis

General Formats:

```
.TRAN{/OP} |print step value| |final time value|
+     {|no print value| {step ceiling value}} {UIC}
```

.TRAN tells PSPICE to run a transient analysis (over time) from **t = 0** to **t = |final time value|**. PSPICE will alter the time step as it simulates - so when there is little going on the time step will jump a lot. This helps speed up PSPICE simulation times. The default maximum time step is **|final time value|**/50. **{step ceiling}** is optional and tells PSPICE the ceiling for the internal time step jump.

|print step value| tells PSPICE the time interval used for printing or plotting information to the **.OUT** file.

|no print value| tells PSPICE not to print info from **t = 0** to **t = |no print value|**.

PSPICE always does a bias point analysis before starting the transient analysis. The optional **{/OP}** tells PSPICE to print the bias point information to the **.OUT** file (similar to the **.OP** statement).

{UIC} tells PSPICE to skip bias point calculations. This option is often used with the **.IC** (initial condition) for capacitors and inductors.

Example:

```
.TRAN 20n 1u
```

tells PSPICE to run a transient analysis from t = 0 to t = 1us. Output is printed (or plotted) to the **.OUT** file every 20ns.

.WIDTH - Width

General Formats:

.WIDTH OUT = <value>

.WIDTH tells PSPICE the character width of the printed output file in characters. The default width is 80 characters.

Example:

.WIDTH OUT = 120
the **.OUT** file will have 120 character-wide lines.

B - GaAsFET

General Formats:

B|name| |drain| |gate| |source| |model| {area value}

B declares a GaAsFET. PSPICE models a GaAsFET as an intrinsic FET with an ohmic resistance (RD/area) in series with the drain, an ohmic resistance (RS/area) in series with the source, and an ohmic resistance (RG) in series with the gate.

{area value} is the relative device area with default = 1.

TABLE A.4. GaAsFET Model Parameters

Parameter	*Description*	*Default Value, Units*
LEVEL	model type (1 = Curtiss, 2 = Raytheon)	Default value = 1
VTO	threshold voltage	Default value = -2.5 ; Units = volt
ALPHA	tanh constant	Default value = 2 ; Units = $volt^{-1}$
B	doping tail extending parameter (level 2 only)	Default value = .3
BETA	transconductance coefficient	Default value = 0.1 ; Units = $amp/volt^2$
LAMBDA	channel-length modulation	Default value = 0 ; Units = $volt^{-1}$
RG	gate ohmic resistance	Default value = 0 ; Units = ohm
RD	drain ohmic resistance	Default value = 0 ; Units = ohm
RS	source ohmic resistance	Default value = 0 ; Units = ohm
IS	gate p-n saturation current	Default value = 1E-14 ; Units = amp
M	gate p-n grading coefficient	Default value = 0.5
N	gate p-n emission coefficient	Default value = 1
VBI	gate p-n potential	Default value = 1 ; Units = volt
CGD	gate-drain zero-bias p-n capacitance	Default value = 0 ; Units = farad
CGS	gate-source zero-bias p-n capacitance	Default value = 0 ; Units = farad
CDS	drain-source zero-bias p-n capacitance	Default value = 0 ; Units = farad
TAU	transit time	Default value = 0 ; Units = sec
FC	forward bias depletion capacitance coefficient	Default value = 0.5
VTOTC	VTO temperature coefficient	Default value = 0 ; Units = volt/°C
BETATCE	BETA exponential temperature coefficient	Default value = 0 ; Units = %/°C
KF	flicker noise coefficient	Default value = 0
AF	flicker noise exponent	Default value = 0

Examples:

B1 100 1 0 MGAAS
declares a GaAsFET **B1** of model **MGAAS**

B2 100 10 0 MGNOM 2.0
declares a GaAsFET **B2** of model **MGNOM** and area multiplier of 2.0

C - Capacitor

General Formats:

C|name| |+ node| |- node| {model name} |value| {IC = |initial value|}

The **|+ node|** and **|- node|** define the polarity of the capacitor. Positive current flows from the **|+ node|** to the **|-node|**.

{model name} is optional and if not included then **|value|** is the capacitance in farads. If **{model name}** is specified then the capacitance is given by:

Ctot = |value| * C * (1+ VC1*V+VC2*V²) * [1+ TC1*(T-Tnom)) + TC2 * (T-Tnom)²]

where **C**, **VC1**, **VC2**, **TC1**, and **TC2** are described below. **Ctot** is the total capacitance. **V** is the voltage across the capacitor. **T** is the simulation temperature. And **Tnom** is the nominal temperature (27°C unless set by **.OPTIONS TNOM** command)

|value| can either be positive or negative.

{IC = |initial value|} gives PSPICE an initial guess for voltage across the capacitor during bias point calculation and is optional.

The capacitor does not have a noise model.

TABLE A.5. Capacitor Model Parameters:

Parameter	*Description*	*Default Value, Units*
C	capacitance multiplier	Default value = 1
TC1	linear temperature coefficient	Default value = 0 ; Units = $°C^{-1}$
TC2	quadratic temperature coefficient	Default value = 0 ; Units = $°C^{-2}$
VC1	linear voltage coefficient	Default value = 0 ; Units = $volt^{-1}$
VC2	quadratic voltage multiplier	Default value = 0 ; Units = $volt^{-2}$

Example:

C1 1 0 20pF
defines a 20pF capacitor between nodes 1 and 0.

D - Diode

General Formats:

D|name| |+ node| |- node| |model name| {area value}

The diode is modeled by a resistor of value R_S/**{area value}** in series with an intrinsic diode. **|+ node|** is the anode and **|- node|** is the cathode. Positive current flows from the anode to cathode.

{area value} scales **IS**, **RS**, **CJO**, and **IBV** and is 1 by default. **IBV** and **BV** are both positive.

Table A.6. Diode Model Parameters:

Parameter	***Description***	***Default Value, Units***
AF	flicker noise exponent	Default value = 1
BV	reverse breakdown value	Default value = infinite ; Units = volt
CJO	zero-bias p-n capacitance	Default value = 0 ; Units = farad
EG	bandgap voltage	Default value = 1.11 ; Units = eV
FC	forward bias depletion capacitance coefficient	Default value = 0.5
IBV	reverse breakdown current	Default value = 1E-10 ; Units = amp
IS	saturation current	Default value = 1E-14 ; Units = amp
KF	flicker noise coefficient	Default value = 0
M	p-n grading coefficient	Default value = 0.5
N	emission coefficient	Default value = 1
RS	parasitic resistance	Default value = 0 ; Units = ohm
TT	transit time	Default value = 0 ; Units = sec
VJ	p-n potential	Default value = 1 ; Units = volt
XTI	IS temperature exponent	Default value = 3

Example:

D1 1 2 DMOD
defines a diode with the characteristics of model **DMOD,** with node **1** as its anode and node **2** as its cathode.

E - Voltage-Controlled Voltage Source

General Formats:

```
E|name| |+ node| |- node|
+       |+ control node| |- control node| |gain|
```

or

```
E|name| |+ node| |- node| POLY(|value|)
+       (|+ control node|, |- control node| ...)
+       |polynomial coefficient value ...|
```

Both formats declare a voltage source whose magnitude is related to the voltage difference between nodes **|+ control node|** and **|- control node|**.

The first form generates a linear relationship. Thus:
Vtot = |gain|*(|+control node|-|- control node|)
where **Vtot** is the voltage between nodes **|+ node|** and **|- node|**.

The second form generates a nonlinear response. The dimension of the polynomial is given by the **|value|**. The dimension means the number of pairs of controlling nodes. See third example below.

In all cases positive current flows from **|+ node|** through the source and out **|- node|**.

Examples:

E1 1 2 3 4 10
gives V(1) - V(2) = 10 * (V(3) - V(4))

E2 5 6 POLY(1) (7,8) 10 20 30
gives V(5) - V(6) = 10 + 20 * (V(7) - V(8)) + 30 * (V(7) - V(8))^2

E3 1 2 POLY(2) (3,4) (5,6) 10 20 30
gives V(1) - V(2) = 10 + 20 * (V(3) - V(4)) + 30 * (V(5) - V(6))

F - Current-Controlled Current Source

General Formats:

```
F|name| |+ node| |- node|
+       |controlling V source| |gain|
```

or

```
F|name| |+ node| |- node| POLY(|value|)
+       (|controlling V source| ...)
+       |polynomial coefficient value ...|
```

Both formats declare a current source whose magnitude is related to the current passing through **|controlling V source|**.

The first form generates a linear relationship. Thus **Itot** = **|gain|*I(|controlling V source|)** where **Itot** is the total current thru the declared **F|name|** device.

The second form generates a nonlinear response. The dimension of the polynomial is given by the **|value|**. The dimension means the number of **|controlling V source|**. See third example below.

In all cases positive current flows from **|+ node|** through the source and out **|- node|**.

Examples:

F1 1 2 VIN 10
gives I(F1) = 10 * I(VIN)

F2 5 6 POLY(1) VIN 10 20 30
gives I(F2) = 10 + 20 * I(VIN) + 30 * (I(VIN)^2)

F3 1 2 POLY(2) VA VB 10 20 30
gives I(F3) = 10 + 20 * I(VA) + 30 * I(VB)

G - Voltage-Controlled Current Source

General Formats:

```
G|name| |+ node| |- node|
+       |+ control node| |- control node| |transconductance|
```

or

```
G|name| |+ node| |- node| POLY(|value|)
+       (|+ control node|, |- control node| ...)
+       |polynomial coefficient value ...|
```

Both formats declare a current source whose magnitude is related to the voltage difference between nodes **|+ control node|** and **|- control node|**.

The first form generates a linear relationship. Thus,

Itot = |transconductance|*(|+control node|-|- control node|)
where **Itot** is the current through declared device **G|name|**.

The second form generates a nonlinear response. The dimension of the polynomial is given by the **|value|**. The dimension means the number of pairs of controlling nodes. See third example below.

In all cases positive current flows from **|+ node|** through the source and out **|- node|**.
Examples:

G1 1 2 3 4 10

gives I(G1) = 10 * (V(3) - V(4))

G2 5 6 POLY(1) (7,8) 10 20 30
gives I(G2) = 10 + 20 * (V(7) - V(8)) + 30 * (V(7) - V(8))^2

G3 1 2 POLY(2) (3,4) (5,6) 10 20 30
gives I(G3) = 10 + 20 * (V(3) - V(4)) + 30 * (V(5) - V(6))

H - Current-Controlled Voltage Source

General Formats:

```
H|name| |+ node| |- node|
+       |controlling V source| |transresistance|
```

or

```
H|name| |+ node| |- node| POLY(|value|)
+       (|controlling V source| ...)
+       |polynomial coefficient value ...|
```

Both formats declare a voltage source whose magnitude is related to the current passing through **|controlling V source|**.

The first form generates a linear relationship. Thus:

Vtot = |transresistance|*I(|controlling V source|)
where **Vtot** is the voltage across **|+ node|** and **|- node|**.

The second form generates a nonlinear response. The dimension of the polynomial is given by the **|value|**. The dimension means the number of **|controlling V source|**. See third example below.

In all cases positive current flows from **|+ node|** through the source and out **|- node|**.

Examples:

H1 1 2 VIN 10
gives V(1) - V(2) = 10 * I(VIN)

H2 5 6 POLY(1) VIN 10 20 30
gives V(5) - V(6) = 10 + 20 * I(VIN) + 30 * (I(VIN)^2)

H3 1 2 POLY(2) VA VB 10 20 30
gives V(1) - V(2) = 10 + 20 * I(VA) + 30 * I(VB)

I - Independent Current Source

General Formats:

I|name| |+ node| |- node| {DC} |value|

or

I|name| |+ node| |- node| AC |magnitude| {phase}

or

I|name| |+ node| |- node| [transient specification]

I declares a current source. There are three types of current sources. **DC**, **AC**, or transient sources.

DC sources give a current source with constant magnitude current. **DC** sources are used for supplies or for .**DC** analyses.

AC sources are used for the **.AC** analysis. The magnitude of the source is given by **|magnitude|**. The initial phase of the source is given by {phase}, default phase is 0.

Transient sources are sources whose output varies over the time of simulation. These are used mostly with the transient analysis, **.TRAN**.

Transient sources must be defined as one of the below:

```
EXP |parameters|
PULSE |parameters|
PWL |parameters|
SFFM |parameters|
SIN |parameters|
```

Positive current flows from **|+ node|** thru the source and out **|- node|**.

These sources can be combined (see second and third examples below).

Examples:

I1 1 2 3mA
declares **I1** as a DC source of magnitude 3mA

I2 3 4 AC 1
declares **I2** an AC source of magnitude 1A (0 initial phase)

I3 5 6 SIN (1 .1 1MEG)
declares **I3** a sinusoidal source with magnitude .1A peak, frequency 1 MHz, and DC offset 1A .

I4 7 8 DC 1 AC 1
declares **I4** as a DC source with magnitude of 1A, with an AC component of magnitude 1A (initial phase = 0)

J - Junction FET

General Formats:

J|name| |drain| |gate| |source| |model| {area}

J declares a JFET. The JFET is modeled as an intrinsic FET with an ohmic resistance (**RD/{area}**) in series with the drain, an ohmic resistance (**RS/{area}**) in series with the source, and an ohmic resistance (**RG**) in series with the gate.

Positive current is defined as flowing into each terminal.

{area}, optional, is the relative device area. It's default is 1.

Example:

J1 1 2 3 MJFET
declares a JFET with drain, gate, source nodes as node **1**, node **2**, node **3** respectively. **MJFET** is the model name of the JFET and must be declared by the **.MODEL** command.

TABLE A.7. JFET Model Parameters

Parameter	*Description*	*Default Value, Units*
AF	flicker noise exponent	Default value = 0
BETA	transconductance coefficient	Default value = 1E-4 ; Units = amp/volt^2
BETATCE	BETA exponential temperature coefficient	Default value = 0 ; Units = %/°C
CGD	gate-drain zero-bias p-n capacitance	Default value = 0 ; Units = farad
CGS	gate-source zero-bias p-n capacitance	Default value = 0 ; Units = farad
FC	forward bias depletion capacitance coefficient	Default value = 0.5
IS	gate p-n saturation current	Default value = 1E-14 ; Units = amp
KF	flicker noise coefficient	Default value = 0
LAMBDA	channel-length modulation	Default value = 0 ; Units = volt^{-1}
RD	drain ohmic resistance	Default value = 0 ; Units = ohm
RS	source ohmic resistance	Default value = 0 ; Units = ohm
VTO	threshold voltage	Default value = -2.0 ; Units = volt
VTOTC	VTO temperature coefficient	Default value = 0 ; Units = volt/°C

K - Inductor Coupling (Transformer Core)

General Formats:

```
K|name| L|inductor name| |L|inductor name|...|
+     |coupling value|
```

or

```
K|name| |L|inductorname|...| |coupling value|
+     |model name| {size value}
```

K couples two or more inductors together. Using the dot convention, place a dot on the first node of each inductor. Then the coupled current will be of opposite polarity with respect to the driving current.

|coupling value| is the coefficient of mutual coupling and must be between 0 and 1. **{size value}** scales the magnetic cross section, it's default is 1.

If **|model name|** is present 4 things change:

1. The mutual coupling inductor becomes a nonlinear magnetic core.
2. The core's B-H characteristics are analyzed using the Jiles-Atherton model.
3. The inductors become windings, thus the number specifying inductance now means number of turns.
4. The list of coupled inductors may just be one inductor.

TABLE A.8. Inductor Coupling Model Parameters

Parameter	*Description*	*Default Value, Units*
A	shape parameter	Default value = 1000 ; Units = amp/meter
ALPHA	mean field parameter	Default value = 0.001
AREA	mean magnetic cross section	Default value = 0.1 ; Units = cm^2
C	domain wall flexing coefficient	Default value = 0.2
GAP	effective air gap length	Default value = 0 ; Units = cm
K	domain wall pinning constant	Default value = 500
MS	magnetization saturation	Default value = 1E6 ; Units = amp/meter
PACK	pack (stacking) factor	Default value = 1
PATH	mean magnetic path length	Default value = 1 ; Units = cm

Example:

```
K1 L1 L2 0.9
```

defines the mutual coupling between inductors **L1** and **L2** is **0.9**. **L1** and **L2** should both be declared somewhere in the net list.

L - Inductor

General Formats:

```
L|name| |+ node| |- node| {model name} |value| {IC = |initial value|}
```

L defines an inductor. **|+ node|** and **|- node|** define the polarity of positive voltage drop. Positive current flows from the **|+ node|** trough the device and out the **|- node|**.

|value| can be positive or negative but not 0.

{model name} is optional. If left out the inductor has an inductance of **|value|** henries.

If **{model name}** is included, then the total inductance is:

$$Ltot=|value| * L * (1+IL1 * I+IL2*I^2) * (1+TC1 * (T-Tnom) + TC2 * (T-Tnom)^2)$$

where **L**, **IL1**, **IL2**, **TC1**, and **TC2** are defined in the model declaration, **T** is the temperature of simulation, and **Tnom** is the nominal temperature (27°C unless specified by **.OPTIONS TNOM**)

{IC = |initial value|} is optional and, if used, defines the initial guess for the current through the inductor when PSPICE attempts to find the bias point.

TABLE A.9. Inductor Model Parameters

Parameter	*Description*	*Default Value, Units*
IL1	linear current coefficient	Default value = 0 ; Units = amp^{-1}
IL1	quadratic current coefficient	Default value = 0 ; Units = amp^{-2}
L	inductance multiplier	Default value = 1
TC1	linear temperature coefficient	Default value = 0 ; Units = °C^{-1}
TC2	quadratic temperature coefficient	Default value = 0 ; Units = °C^{-2}

Example:

```
L1 1 2 10m
```
defines an inductor between nodes 1 and 2 with inductance of 10mH.

M - MOSFET

General Format:

```
M|name| |drain| |gate| |source| |substrate| |model name|
+     {L = |value|} {W = |value|} {AD = |value|} {AS = |value|}
+     {PD = |value|} {PS = |value|} {NRD = |value|} {NRS = |value|}
+     {NRG = |value|} {NRB = |value|}
```

M defines a MOSFET transistor. The MOSFET is modeled as an intrinsic MOSFET with ohmic resistances in series with the drain, source, gate, and substrate(bulk). There is also a shunt resistor (**RDS**) in parallel with the drain-source channel. Positive current is defined to flow into each terminal.

L and **W** are the channel's length and width. **L** is decreased by **2*LD** and **W** is decreased by **2*WD** to get the effective channel length and width. **L** and **W** can be defined in the device statement, in the model, or in **.OPTION** command. The device statement has precedence over the model which has precedence over the **.OPTIONS**.

AD and **AS** are the drain and source diffusion areas. **PD** and **PS** are the drain and source diffusion parameters. The drain-bulk and source-bulk saturation currents can be specified by **JS** (which in turn is multiplied by **AD** and **AS**) or by **IS** (an absolute value). The zero-bias depletion capacitances can be specified by **CJ**, which is multiplied by **AD** and **AS**, and by **CJSW**, which is multiplied by **PD** and **PS**, or by **CBD** and **CBS**, which are absolute values. **NRD**, **NRS**, **NRG**, and **NRB** are reactive resistivities of their respective terminals in squares. These parasitics can be specified either by **RSH** (which in turn is multiplied by **NRD**, **NRS**, **NRG**, and **NRB**) or by absolute resistances **RD**, **RG**, **RS**, and **RB**. Defaults for **L**, **W**, **AD**, and **AS** may be set using the **.OPTIONS** command. If **.OPTIONS** is not used their default values are 100u, 100u, 0, and 0 respectively

TABLE A.10. MOSFET Model Parameters:

Parameter	*Description*	*Default Value, Units*
AF	Flicker noise exponent	Default value = 1
CBD	bulk-drain zero-bias p-n capacitance	Default value = 0 ; Units = farad
CBS	bulk-source zero-bias p-n capacitance	Default value = 0 ; Units = farad
CGBO	gate-substrate overlap capacitance/channel length	Default value = 0 ; Units = farad/meter
CGDO	gate-drain overlap capacitance/channel width	Default value = 0 ; Units = farad/meter
CGSO	gate-source overlap capacitance/channel width	Default value = 0 ; Units = farad/meter
CJ	bulk p-n zero-bias bottom capacitance/area	Default value = 0 ; Units = farad/meter^2
CJSW	bulk p-n zero-bias bottom capacitance/area	Default value = 0 ; Units = farad/meter^2
DELTA	width effect on threshold	Default value = 0
ETA	static feedback (LEVEL = 3)	Default value = 0
FC	bulk p-n forward-bias capacitance coefficient	Default value = 0.5
GAMMA	bulk threshold parameter	Default value = 0 ; Units = volt$^{0.5}$
IS	bulk p-n saturation current	Default value = 1E-14 ; Units = amp
JS	bulk p-n saturation current/area	Default value = 0 ; Units = amp/meter^2
KAPPA	saturation field factor (LEVEL = 3)	Default value = 0.2
KF	Flicker noise coefficient	Default value = 0
KP	transconductance	Default value = 2E-5 ; Units = amp/volt^2
L	channel length	Default value = DEFL ; Units = meter
LAMBDA	channel length modulation (LEVEL = 1 or 2)	Default value = 0 ; Units = volt^{-1}
LD	lateral diffusion (length)	Default value = 0 ; Units = meter
LEVEL	model type (1, 2, or 3)	Default value = 1
MJ	bulk p-n bottom grading coefficient	Default value = 0.5
MJSW	bulk p-n sidewall grading coefficient	Default value = 0.33
NEFF	channel charge coefficient (LEVEL = 2)	Default value = 1
NFS	fast surface state density	Default value = 0 ; Units = 1/cm^2
NSS	surface state density	Default value = 0 ; Units = 1/cm^2
NSUB	substrate doping density	Default value = 0 ; Units = 1/cm^3
PB	bulk p-n potential	Default value = 0.8 ; Units = volt
PHI	surface potential	Default value = 0.7 ; Units = volt
RB	substrate ohmic resistance	Default value = 0 ; Units = ohm
RD	drain ohmic resistance	Default value = 0 ; Units = ohm
RDS	drain-source ohmic resistance	Default value = infinite ; Units = ohm
RG	gate ohmic resistance	Default value = 0 ; Units = ohm
RS	source ohmic resistance	Default value = 0 ; Units = ohm
RSH	drain, source diffusion sheet resistance	Default value = 0 ; Units = ohm/square
THETA	mobility modulation (LEVEL = 3)	Default value = 0 ; Units = volt^{-1}
TOX	oxide thickness	Default value = infinite ; Units = meter
TPG	gate material type:+1 = opposite, -1 = same, 0 = aluminum	Default value = +1
UCRIT	mobility degradation critical field (LEVEL = 2)	Default value = 1E4 ; Units = volt/cm
UEXP	mobility degradation exponent (LEVEL =2)	Default value = 0
UO	surface mobility	Default value = 600 ; Units = cm^2/(volt*sec)
VMAX	maximum drift velocity	Default value = 0; Units = meter/sec
VTO	zero-bias threshold voltage	Default value = 0 ; Units = volt
W	channel width	Default value = DEFW ; Units = meter
WD	lateral diffusion (width)	Default value = 0 ; Units = meter
XJ	metallurgical junction depth	Default value = 0 ; Units = meter
XQC	fraction of channel charge attributed to drain	Default value = 1

Examples:

`M1 1 2 3 0 MNMOS L=3u W=1u`

defines a MOSFET with drain node 1, gate node 2, source node 3, substrate node 0, channel length and width 3u and 1u respectively, and described further by model **`MNMOS`** (which is assumed to exist in the **`.MODEL`** statements)

`M2 4 5 6 0 MNMOS`

defines a MOSFET with drain node 4, gate node 5, source node 6, substrate node 0, and described further by model **`MNMOS`** (which is assumed to exist in the **`.MODEL`** statements)

Q - Bipolar Transistor

General Formats:

```
Q|name| |collector| |base| |emitter|
+     {substrate} |model name| {area value}
```

`Q` declares a bipolar transistor in PSPICE. The transistor is modeled as an intrinsic transisitor with ohmic resistances in series with the base, the collector (**`RC/{area value}`**), and with the emitter (**`RE/{area value}`**). **`{substrate}`** node is optional, default value is ground. Positive current is defined as flowing into a terminal. **`{area value}`** is optional (used to scale devices), default is 1. The parameters **`ISE`** and **`ISC`** may be set greater than 1. If so they become multipliers of **`IS`** (i.e. **`ISE*IS`**).

TABLE A.11. Bipolar Transistor Model Parameters

Parameter	*Description*	*Default Value, Units*
AF	Flicker noise exponent	Default value = 1
BF	ideal maximum forward beta	Default value = 100
BR	ideal maximum reverse beta	Default value = 1
CJC	base-collector zero-bias p-n capacitance	Default value = 0; Units = farad
CJE	base-emitter zero-bias p-n capacitance	Default value = 0; Units = farad
CJS	collector-substrate zero-bias p-n capacitance	Default value = 0; Units = farad
EG	bandgap voltage (barrier height)	Default value = 1.11 ; Units = eV
FC	forward bias depletion capacitor coefficient	Default value = 0.5
IKF	corner for forward beta high current roll off	Default value = infinite ; Units = amp
IKR	corner for reverse beta high current roll off	Default value = infinite ; Units = amp
IRB	current at which RB falls halfway to RBM	Default value = infinite ; Units = amp
IS	p-n saturation current	Default value = 1E-16 ; Units = amp
ISC	base-collector leakage saturation coefficient	Default value = 0 ; Units = amp
ISE	base-emitter leakage saturation current	Default value = 0 ; Units = amp
ITF	Transit time dependency on IC	Default value = 0 ; Units = amp
KF	Flicker noise coefficient	Default value = 0
MJC	base-collector p-n grading coefficient	Default value = 0.33
MJE	base-emitter p-n grading coefficient	Default value = 0.33
MJS	collector-substrate p-n grading coefficient	Default value = 0
NC	base-collector leakage emission coefficient	Default value = 2.0
NE	base-emitter leakage emission coefficient	Default value = 1.5
NF	forward current emission coefficient	Default value = 1
NR	reverse current emission coefficient	Default value = 1
PTF	excess phase at 1/(2*PI*TF) Hz.	Default value = 0 ; Units = degree
RB	zero-bias (maximum) base resistance	Default value = 0 ; Units = ohm
RBM	minimum base resistance	Default value = RB ; Units = ohm
RC	collector ohmic resistance	Default value = 0 ; Units = amp
RE	emitter ohmic resistance	Default value = 0 ; Units = ohm
TF	ideal forward transit time	Default value = 0 ; Units = sec
TR	ideal reverse transit time	Default value = 0 ; Units = sec
VAF	forward Early voltage	Default value = infinite ; Units = volt
VAR	reverse Early voltage	Default value = infinite ; Units = volt
VJC	base-collector built in potential	Default value = 0.75 ; Units = volt
VJE	base-emitter built in potential	Default value = 0.75 ; Units = volt
VJS	collector-substrate built in potential	Default value = 0.75 ; Units = volt
VTF	transit time dependency on VBC	Default value = infinite ; Units = volt
XCJC	fraction of CJC connected internal to RB	Default value = 1
XTB	forward and reverse bias temperature coefficient	Default value = 0
XTF	transit time bias dependence coefficient	Default value = 0
XTI	IS temperature effect exponent	Default value = 3

Example:

Q1 1 2 3 MNPN
defines a bipolar transistor of model MNPN with collector, base, and emitter nodes of 1, 2, and 3 respectively.

R - Resistor

General Formats:

R|name| |+ node| |- node| {model name} |value|

The **|+ node|** and **|- node|** define the polarity of the resistor in terms of the voltage drop across it. Positive current flows from the **|+ node|** through the resistor and out the **|- node|**.

{model name} is optional and if not included then |value| is the resistance in ohms. If **{model name}** is specified and **TCE** is not specified then the resistance is given by:

Rtot = |value| * R * [1+TC1 * (T-Tnom)) + TC2 * (T-Tnom)^2]

where **R**, **TC1**, and **TC2** are described below. **Rtot** is the total resistance. **V** is the voltage across the resistor. **T** is the simulation temperature. And **Tnom** is the nominal temperature (27°C unless set by **.OPTIONS TNOM** command)

If **TCE** is specified then the resistance is given by:

Rtot = |value| * R * 1.01$^{(TCE*(T-Tnom))}$

|value| can either be positive or negative.

Noise is calculated using a 1 Hz bandwidth. The resistor generates thermal noise with the following spectral power density (per unit BW):

i^2 = 4 * k * T/resistance

where **k** is Boltzmann's constant.

TABLE A.12. Resistor Model Parameters:

Parameter	***Description***	***Default Value, Units***
R	resistance multiplier	Default value = 1
TC1	linear temperature coefficient	Default value = 0 ; Units = °C^{-1}
TC2	quadratic temperature coefficient	Default value = 0 ; Units = °C^{-2}
TCE	exponential temperature coefficient	Default value = 0 ; Units = %/°C

Example:

R1 1 0 20
defines a 20 ohm resistor between nodes 1 and 0.

S - Voltage-Controlled Switch

General Formats:

```
S|name| |+ switch node| |- switch node|
+     |+ control node| |- control node| |model name|
```

S denotes a voltage controlled switch. The resistance between **|+ switch node|** and **|- switch node|** depends on the voltage difference between **|+ control node|** and **|- control node|**. The resistance varies continuously between **RON** and **ROFF**.

RON and **ROFF** must be greater than zero and less than **GMIN** (set in the **.OPTIONS** command). A resistor of value **1/GMIN** is connected between the controlling nodes to prevent them from floating.

TABLE A.13. Voltage Controlled Switch Model Parameters:

Parameter	*Description*	*Default Value, Units*
RON	on resistance	Default value = 1 ; Units = ohm
ROFF	off resistance	Default value = 1E6 ; Units = ohm
VON	control voltage for on state	Default value = 1 ; Units = volt
VOFF	control voltage for off state	Default value = 0 ; Units = volt

Example:

```
S1 1 2 3 4  MSW
```

defines a current controlled switch. The resistance between nodes 1 and 2 varies with the voltage difference between nodes 3 and 4. The switch model is **MSW**.

T - Transmission Line

General Formats:

```
T|name| |+ A port| |- A port| |+ B port| |- B port|
+     Z0 = |value| {TD = |TD value|} {F = |F value|{NL = |NL value|}}
```

T defines a 2 port transmission line. The device is a bi-directional, ideal delay line. The two ports are **A** and **B** with their polarities given by the **+** or **-** sign. **Z0** is the characteristic impedance of the line.

The length of the transmission line can either be defined by **TD**, the delay in seconds, or by **F** and **NL** a frequency and relative wavelength.

Example:

```
T1 1 0 2 0 Z0=50 F=1E9 NL=0.25
```

declares a two port transmission line. The two ports are given as nodes 1 and 2. The line has a characteristic impedance of 50 ohms and a length of 0.25 wavelengths at 1 GHz.

V - Independent Voltage Source

General Formats:

```
V|name| |+ node| |- node| {DC} |value|
```

or

```
V|name| |+ node| |- node| AC |magnitude| {phase}
```

or

```
V|name| |+ node| |- node| [transient specification]
```

V declares a voltage source. There are three types of voltage sources. **DC**, **AC**, or transient sources.

DC sources give a voltage source with constant magnitude voltage. **DC** sources are used for supplies or for **.DC** analyses.

AC sources are used for the **.AC** analysis. The magnitude of the source is given by **|magnitude|**. The initial phase of the source is given by **{phase}**, default phase is 0.

Transient sources are sources whose output varies over the time of simulation. These are used mostly with the transient analysis, **.TRAN**.

Transient sources must be defined as one of the below:

```
EXP |parameters|
PULSE |parameters|
PWL |parameters|
SFFM |parameters|
SIN |parameters|
```

Positive current flows from **|+ node|** through the source and out **|- node|**. These sources can be combined (see second and third examples below).

Examples:

V1 1 2 1
declares **V1** as a DC source of magnitude 1 V.

V2 3 4 AC 1
declares **V2** an AC source of magnitude 1 V (0 initial phase)

V3 5 6 SIN (1 .1 1MEG)
declares **V3** a sinusoidal source with magnitude .1 V peak, frequency 1 MHz, and DC offset 1 V .

V4 7 8 DC 1 AC 1
declares V4 as a DC source with magnitude of 1 V,with an AC component of magnitude 1 V (initial phase = 0)

W - Current-Controlled Switch

General Formats:

```
W|name| |+ switch node| |- switch node|
+     |controlling V source| |model name|
```

W denotes a current controlled switch. The resistance between **|+ switch node|** and **|- switch node|** depends on the current flowing through the control source **|controlling V source|**. The resistance varies continuously between **RON** and **ROFF**.

RON and **ROFF** must be greater than zero and less than **GMIN** (set in the **.OPTIONS** command). A resistor of value 1/GMIN is connected between the controlling nodes to prevent them from floating.

TABLE A.14. Current-Controlled Switch Model Parameters:

Parameter	***Description***	***Default Value, Units***
RON	on resistance	Default value = 1 ; Units = ohm
ROFF	off resistance	Default value = 1E6 ; Units = ohm
ION	control voltage for on state	Default value = 0.001 ; Units = amp
IOFF	control voltage for off state	Default value = 0 ; Units = amp

Example:

W1 1 2 VCONT MSW
defines a current controlled switch. The resistance between nodes 1 and 2 varies with the current flowing through the control source **VCONT**. The switch model is **MSW**.

X - Sub-circuit Call

General Formats:

X|name| {node ...} |Sub-circuit name|

X calls the Sub-circuit **|Sub-circuit name|**. **|Sub-circuit name|** must somewhere be defined by the **.SUBCKT** and **.ENDS** command. The number of nodes (given by **{node ...}**) must be consistent. The referenced Sub-circuit is inserted into the given circuit with the given nodes replacing the argument nodes in the definition. Sub-circuit calls may be nested but cannot become circular.

Example:

X1 1 2 OPAMP
calls the Sub-circuit **OPAMP**.

SOURCES - Transient Source Descriptions

There are several types of available sources for transient declarations. Each kind, its description, and an example is given below:

EXP - Exponential Source

General Format:

EXP (|v1| |v2| |td1| |td2| |tau1| |tau2|)

The **EXP** form causes the voltage to be **|v1|** for the first **|td1|** seconds. Then it grows exponentially from **|v1|** to **|v2|** with time constant **|tau1|**. The growth lasts **|td2|** - **|td1|** seconds. Then the voltage decays from **|v2|** to **|v1|** with time constant **|tau2|**.

TABLE A.15. Exponential Source Model Parameters

Parameter	*Description*	*Default Value, Units*
v1	initial voltage	Default value = none ; Units = volt
v2	peak voltage	Default value = none ; Units = volt
td1	rise delay time	Default value = 0 ; Units = second
tau1	rise time constant	Default value = **TSTEP** ; Units = second
td2	fall delay time	Default value = **td1 + TSTEP** ; Units = second
tau2	fall time constant	Default value = **TSTEP** ; Units = second

Example:

V1 1 0 EXP(0 1 2u 10u 50u 10u)

PULSE - Pulse source

General Format:

PULSE(|v1| |v2| |td| |tr| |tf| |pw| |per|)

Pulse generates a voltage to start at **|v1|** and hold there for **|td|** seconds. Then the voltage goes linearly from **|v1|** to **|v2|** for the next **|tr|** seconds. The voltage is then held at **|v2|** for **|pw|** seconds. Afterwards, it changes linearly from **|v2|** to **|v1|** in **|tf|** seconds. It stays at **|v1|** for the remainder of the period given by **|per|**.

TABLE A.16. Pulse Source Model Parameters

Parameter	*Description*	*Default value, Units*
v1	initial voltage	Default value = none ; Units = volt

v2	pulsed voltage	Default value = none ; Units = volt
td	delay time	Default value = 0 ; Units = second
tr	rise time	Default value = TSTEP ; Units = second
tf	fall time	Default value = TSTEP ; Units = second
pw	pulse width	Default value = TSTOP ; Units = second
per	period	Default value = TSTOP ; Units = second

Example:

```
V1 1 0 PULSE(0 5 2u 10u 10u 100u 300u)
```

PWL - Piecewise Linear Source

General Format:

```
PWL(|t1| |v1| |t2| |v2| .... |ti| |vi| )
```

PWL describes a piecewise linear format. Each pair of time/voltage (i.e. **|t1|**, **|v1|**) specifies a corner of the waveform. The voltage between corners is the linear interpolation of the voltages at the corners.

TABLE A.17. Piecewise-Linear Source Model Parameters

Parameter	*Description*	*Default Value, Units*
ti	corner time	Default value = none ; Units = second
vi	corner voltage	Default value = none ; Units = volt

Example:

```
V1 1 0 PWL(0 0 1u 0 1.01u 5 10m 5)
```

SFFM - Single Frequency FM Source

General Format:

```
SFFM(|vo| |va| |fc| |mdi| |fs|)
```

SFFM causes the voltage signal to follow:

```
v = vo + va * sin(2π * fc * t + mdi * sin(2π * fs * t))
```

where **vo**, **va**, **fc**, **mdi**, and **fs** are defined below. **t** is time.

TABLE A.18. Single Frequency FM Source Model Parameters

Parameter	*Description*	*Default Value, Units*
vo	offset voltage	Default value = none ; Units = volt
va	peak amplitude voltage	Default value = none ; Units = volt
fc	carrier frequency	Default value = 1/TSTOP ; Units = Hz
mdi	modulation index	Default value = 0
fs	signal frequency	Default value = 1/TSTOP ; Units = Hz

Example:

```
v1 1 0 SFFM(3 1 88MEG 0.5 20k)
```

SIN - Sinusoidal Source

General Format:

```
SIN(|vo| |va| |freq| |td| |df| |phase|)
```

SIN creates a sinusoidal source. The signal holds at **|vo|** for **|td|** seconds. Then the voltage becomes an exponentially damped sine wave described by:

$$v = vo + va * \sin(2\pi * (freq * (t - td) - phase/360)) * e^{-((t - td) * df)}$$

TABLE A.19. Sinusoidal Source Model Parameters

Parameter	*Description*	*Default Value, Units*
vo	offset voltage	Default value = none ; Units = volt
va	peak amplitude voltage	Default value = none ; Units = volt
freq	carrier frequency	Default value = 1/TSTOP ; Units = Hz
td	delay	Default value = 0 ; Units = second
df	damping factor	Default value = 0 ; Units = $second^{-1}$
phase	phase	Default value = 0 ; Units = degree

Example:

```
V1 1 0 SIN(2 1 20k 1m 90)
```

Note, that for all of the above, a time varying current source is described as above except interchange current for voltage.